Soluciones a los ejercicios de Prácticas

Ejercicio de curvas

Para una curva genérica en 2-D, y sin pasar por las cuentas en 3-D, calcular su evoluta y dibujarla en 2D junto a la curva que elijas como ejemplo.

Solución.

Hemos elegido aquí la curva pez α(t)=(b(cos(t)-$\frac{\text{sen}(t)^2}{\sqrt{2}}$), b sen(t)cos(t)) con b=1. La introducimos y la dibujamos.

In[88]:=
```
b = 1;
(*Fish curve*)
pez := {b (Cos[t] - Sin[t]^2 / Sqrt[2]), b * Sin[t] * Cos[t]};
ParametricPlot[pez, {t, 0, 2 Pi}]
```

Out[90]=



Para dar la fórmula de la evoluta, sólo necesitamos el normal y la curvatura en cada punto, y para esto necesitamos el tangente.
(omitimos aquí la salida, puesto que su expresión no es lo que nos interesa en este ejemplo)

In[91]:=
```
tang = D[pez, t] / Sqrt[D[pez, t].D[pez, t]] // Simplify;
normal = D[tang, t] / Sqrt[D[tang, t].D[tang, t]] // Simplify;
curvat = Sqrt[D[tang, t].D[tang, t]] / Sqrt[D[pez, t].D[pez, t]] // Simplify;
evoluta = pez + normal / curvat // Simplify
```

Y ya sólo queda dibujar (curva pez en rojo y evoluta en azul):

In[95]:=
```
ParametricPlot[Evaluate[{pez, evoluta}],
 {t, 0, 2 Pi}, PlotStyle -> {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```

Out[95]=



Ejercicio de superficies

a) Determina una parametrización para la superficie que resulta de rotar la curva

$$\alpha(t) = (2+\cos(t), \text{sen}(t)\ , t)$$

en torno al eje OZ.
b) Dibújala
c) Verifica que una de sus curvaturas principales no cambia de signo, y la otra sí.
d) Determina un punto elíptico, uno parabólico, uno hiperbólico y uno umbilical.

Apartado a)

```
In[96]:= Clear["Global`*"]
alpha = {2 + Cos[v], Sin[v], v};
radio = Sqrt[alpha[[1]]^2 + alpha[[2]]^2] // Simplify;
X = {radio * Cos[u], radio * Sin[u], alpha[[3]]};
Print["La parametrización es X(u,v) = ", X]
```

La parametrización es X(u,v) = $\left\{\text{Cos}[u]\sqrt{5+4\,\text{Cos}[v]}\ ,\ \sqrt{5+4\,\text{Cos}[v]}\ \text{Sin}[u],\ v\right\}$

Apartado b)

```
ParametricPlot3D[Evaluate[X], {u, 0, 2 Pi}, {v, 0, 12}]
```

Out[101]=



Apartado c)
Calculamos la matriz de (-DN) (pasando por las matrices de la primera y segunda forma fundamental) y observamos que, en este caso, es diagonal y sólo depende de v.

```
In[106]:= matriz1 =
    {{D[X, u].D[X, u], D[X, u].D[X, v]}, {D[X, u].D[X, v], D[X, v].D[X, v]}} // Simplify;
NN = Cross[D[X, u], D[X, v]] / Sqrt[Cross[D[X, u], D[X, v]].Cross[D[X, u], D[X, v]]] //
    Simplify;
matrizDN = Inverse[-matriz1].{{NN.D[X, {u, 2}], NN.D[X, u, v]},
      {NN.D[X, u, v], NN.D[X, {v, 2}]}} // Simplify;

Print["Matriz de -DN(u,v) = ", matrizDN // MatrixForm]
```

Matriz de -DN(u,v) = $\begin{pmatrix} \frac{1}{\sqrt{7+4\,\text{Cos}[v]-2\,\text{Cos}[2\,v]}} & 0 \\ 0 & \frac{2\,(3+5\,\text{Cos}[v]+\text{Cos}[2\,v])}{(7+4\,\text{Cos}[v]-2\,\text{Cos}[2\,v])^{3/2}} \end{pmatrix}$

Como hemos adelantado, es diagonal, por lo que las curvaturas principales k_1 y k_2 son exactamente los elementos de su diagonal (cambiados de signo). Y como sólo depende de v, los signos de k1 y k2 los observamos al dibujar estas funciones respecto de v: (aunque es este caso, se ve a simple vista que la primera $\frac{-1}{\sqrt{7+4\,\mathrm{Cos}[v]-2\,\mathrm{Cos}[2\,v]}}$ es siempre negativa.

```
k1 = -matrizDN[[1, 1]];
k2 = -matrizDN[[2, 2]];
Plot[{k1, k2}, {v, 0, 12}, PlotRange → {-1, 2}]
```

Out[114]=



Apartado d)

Sólo mirando esta gráfica, observamos varias cosas:

1. Una curvatura siempre es negativa (respuesta al apartado c)
2. Hay valores de v en los que una curvatura se anula (darán puntos parabólicos)
3. Hay valores de v para los que se cortan ambas curvaturas (darán puntos umbilicales)
4. Hay valores de v en los que ambas son negativas (darán puntos elípticos)
5. Hay valores de v en los que una es negativa y la otra positiva (darán puntos hiperbólicos).

Concretando: para v = $\pi/2$ (y u cualquiera), tenemos un punto elíptico y para v = 3 (u cualquiera) uno hiperbólico.
Para los otros dos, resolvemos respectivamente, k2 = 0 y k1 = k2.

In[139]:=
```
parabolico = Solve[k2 == 0, v][[1]]
umbilical = Solve[k1 == k2, v][[1]]
```

Out[139]= $\left\{v \to -\frac{2\pi}{3}\right\}$

Out[140]= $\left\{v \to -\frac{\pi}{3}\right\}$

Y juntando todo, presentamos los puntos pedidos de forma elegante:

In[141]:=
```
Print["Punto elíptico P = ", X /. {u → 0, v → Pi / 2},
 " con curvaturas principales ", k1 /. {u → 0, v → Pi / 2}, " y ", k2 /. {u → 0, v → Pi / 2}]
Print["Punto hiperbólico P = ", X /. {u → 0, v → Pi},
 " con curvaturas principales ", k1 /. {u → 0, v → Pi}, " y ", k2 /. {u → 0, v → Pi}]
Print["Punto parabólico P = ", X /. {u → 0} /. parabolico, " con curvaturas principales ",
 k1 /. {u → 0} /. parabolico, " y ", k2 /. {u → 0} /. parabolico]
Print["Punto umbilical P = ", X /. {u → 0} /. umbilical, " con curvaturas principales ",
 (k1 /. {u → 0} /. umbilical) // Simplify, " y ", (k2 /. {u → 0} /. umbilical) // Simplify]
```

Punto elíptico P = $\left\{\sqrt{5}, 0, \frac{\pi}{2}\right\}$ con curvaturas principales $-\frac{1}{3}$ y $-\frac{4}{27}$

Punto hiperbólico P = $\{1, 0, \pi\}$ con curvaturas principales -1 y 2

Punto parabólico P = $\left\{\sqrt{3}, 0, -\frac{2\pi}{3}\right\}$ con curvaturas principales $-\frac{1}{\sqrt{6}}$ y 0

Punto umbilical P = $\left\{\sqrt{7}, 0, -\frac{\pi}{3}\right\}$ con curvaturas principales $-\frac{1}{\sqrt{10}}$ y $-\frac{1}{\sqrt{10}}$